

CHAPTER 7

USING GRAPHICS AND COLOR

- Graphics characters
- Character animation
- Controlling colors
- High resolution graphics
- Points, lines, and labels
- Squares, circles, polygons, and painting
- Multi-color graphics

GRAPHICS CHARACTERS

Each letter key contains 2 different graphic characters, as do the @, -, *, and English Pound keys. To print graphics characters, you must hold down the **SHIFT** or **CTRL** keys while you press the key for the graphics symbol you want.

When your Plus/4 is in upper-case/graphics mode, hold down **SHIFT** and press a letter key to display the graphics character on the right side of that letter key. These characters include the playing card suits, a solid and a hollow ball, and a set of lines and connecting characters that let you draw many different pictures on your screen.

Here are some examples to help you get used to these characters:

Exercise 1: Large Circle

- Step 1: Press down the **SHIFT LOCK** key.
- 2: Press the letter U then the letter I.
- 3: Press the **RETURN** key.
- 4: Press the letter J then the letter K.
- 5: Press the **RETURN** key.

Exercise 2: Snake

- Step 1: Press down the **SHIFT LOCK** key.
- 2: Press U, then I, then U, then I, then U, then I.
- 3: Press the **RETURN** key.
- 4: Press K, then J, then K, then J, then K, then J.
- 5: Press the **RETURN** key.

Exercise 3: Crooked Line

- Step 1: Press down the **SHIFT LOCK** key.
- 2: Press E, then D, then C, then *, then F, then R.
- 3: Press the **RETURN** key.

Exercise 4: Two Crosses

- Step 1: Press down the **SHIFT LOCK** key.
- 2: Press M, then SPACE, then N, then SPACE, then —.
- 3: Press the **RETURN** key.
- 4: Press SPACE, then V, then SPACE, then *, then +, then *.
- 5: Press the **RETURN** key.
- 6: Press N, then SPACE, then M, then SPACE, then —.
- 7: Press the **RETURN** key.

When you are finished, press [] again so it pops up.

Did you wonder why the computer doesn't say SYNTAX ERROR when you hit []? After all, you had characters on the line that weren't commands that the computer can understand.

The reason is that the Plus/4 doesn't pay attention to the line you typed if you hold down [] when you press []. If you press [] without the [] key, the computer tries to figure out what you mean when you're just drawing pictures.

So far we haven't talked about the graphics characters on the left side of the keys. These graphics work just like the right side characters, except that you hold down the [] key instead of []. There is no [] lock, so you must hold it down yourself.

You can print this set of graphics in either upper-case/graphics mode or upper/lower-case mode. They are always available.

The left side graphics characters include lines and angles used for drawing charts and tables. For example, here's how to underline a word:

First, move the cursor to the line below the word you want to underline. Then hold down the [] key and the T key, which prints an underline graphic. Hold these two keys down until the word is underlined.

Exercise 5: Half Bar

- Step 1: Hold down the [] key with one hand during the whole exercise.
 - 2: Press D, then I, then I, then F.
 - 3: Press the [] key.
-

Exercise 6: Wedge

- Step 1: Hold down the [] key and hit T, then Y, then U.
 - 2: Hold down the [] key and hit 9.
 - 3: Hold down the [] key and hit I, then O, then P, then @, then SPACE.
 - 4: Press the [] key.
-

Exercise 7: Window

- Step 1: Hold down the [] key until step 4.
 - 2: Press A, then R, then S, then [RETURN]
-

- 3: Press Q.
- 4: Let go of **HOME** (it's OK, honest).
- 5: Hold down **SHIFT** and hit +.
- 6: Let go of **SHIFT** and hold down **Q** again. Don't let go of it any more.
- 7: Press W, then **RETURN**.
- 8: Press Z, then E, then X, then **RETURN**.

The purpose of these exercises is to show you how the graphic symbols of the Plus/4 can be manipulated to create different shapes and figures. These are only a handful of the figures and representations you can develop. Now that you have a good idea of what is involved in using the graphic symbols to build different forms, you should experiment with them yourself, and see what you come up with.

CHARACTER ANIMATION

Movies are really a sequence of still pictures. Each picture is a little different from the one that came before. The projector shows each picture for a very short time, then goes on to the next one. The scene becomes animated.

Computer animation works the same way. First the computer draws one picture, then it changes the picture slightly. The Plus/4 is fast enough to allow objects to move smoothly all around the screen in your games and practical programs.

You can't type fast enough to create animation. A movie is animated at a rate of 30 pictures per second. The changes must be fast enough to fool the eye. So you must use a program to draw a picture, wait for a split second, then change to a new picture.

To get the program to create pictures we use the PRINT statement with the graphic characters. The simplest type of animation involves alternating two characters to get the effect of movement.

This program simulates the movement of a pulsing ball.

Type NEW and press **RETURN** before entering each new program. Remember to press **RETURN** after each line in all these programs.

```
10 PRINT " HOME SHIFT Q "
20 FOR L=1 TO 100
30 NEXT L
40 PRINT " HOME SHIFT W "
50 FOR L=1 TO 100
60 NEXT L
70 GOTO 10
```

Type these keys simultaneously

Type RUN and press **RETURN**

- To get a more interesting effect you can build a small picture from several graphic characters, then change a few of the characters while leaving others in the same place. This gives the effect of part of an object moving, demonstrated in the following program.

IMPORTANT TO NOTE: Each time **SHIFT** or **C** is referred to, it should be typed at the SAME TIME as the key following it when entering the program.

```
10 PRINT " HOME SHIFT M SHIFT W SHIFT
N "
20 PRINT "SPACE C + SPACE "
30 PRINT " SHIFT N SPACE SHIFT M "
40 FOR L=1 TO 100: NEXT L
50 PRINT " HOME SPACE SHIFT WSPACE"
60 PRINT " C T C + C T "

70 PRINT "SPACE C G C G "
80 FOR L=1 TO 100: NEXT L
90 GOTO 10
```

Type RUN and press **RETURN**

In both examples of animation so far, we've worked on only one area on the screen. The next step is to move the animated figure around. The TAB function helps when you want to move objects from the left edge. The following program portrays a snake crawling on the screen.

Remember that **SHIFT** and the following key are still typed together.

```
5 FOR A = 0 TO 30
10 PRINT " SHIFT CLR "
20 PRINT TAB (A) " SHIFT U SHIFT I SHIFT U
SHIFT I "
30 PRINT TAB (A) " SHIFT K SHIFT J SHIFT K
SHIFT J "
40 FOR L=1 TO 100: NEXT L
50 PRINT " SHIFT CLR "
60 PRINT TAB (A+1) " SHIFT I SHIFT U SHIFT I
SHIFT U "
70 PRINT TAB (A+1) " SHIFT J SHIFT K SHIFT K
SHIFT J "
80 FOR L=1 TO 100: NEXT L
90 NEXT A
```

Using characters like the ball ( Q), you can play video games on the screen. To move a ball, just erase the ball and replace it at a new position, as in this program.

```
10 PRINT "   "
20 PRINT "   Q  ";
30 FOR L=1 TO 50: NEXT L
40 GOTO 20
```



Left cursor arrow

Type RUN and press the  key. Press the STOP key when you want to stop moving the ball.

CONTROLLING COLORS

Separate colors can be put into each part of the screen. The border can be one color, the background a different one, and each character can have its own color. You already know how to set the character colors using the keyboard. The COLOR command adjusts the color of the other screen areas.

Turn the border of your screen red by typing the command COLOR 4,3 and pressing the  key. The number 4 in the command stands for the border area, and color number 3 is red (the same number as on the key marked RED).

Now type COLOR 0, 7 and hit  . The screen background turns blue. The number 0 stands for the background, while the 7 is blue (also the same as the keyboard).

The first number after the word COLOR stands for the area on the screen you want to change. Area 0 is the background, 1 is the character color, 4 is the border. You'll learn about areas 2 and 3 when you get into multi-color graphics later in this chapter.

Screen Area Numbers

AREA #	AREA NAME
0	Background
1	Character
2	Multi-color 1
3	Multi-color 2
4	Border

Each color also has an adjustable brightness level, called the *luminance*. You can add a number from 0 (darkest) through 7 (brightest) after the color number to vary the color. Type COLOR 4 ,3, 0 and hit  . The border becomes a dark red. Type COLOR 4, 3, 7 and the border changes to a bright red.

Color Numbers

COLOR#	COLOR	COLOR#	COLOR
1	BLACK	9	ORANGE
2	WHITE	10	BROWN
3	RED	11	YELLOW GREEN
4	CYAN	12	PINK
5	PURPLE	13	BLUE GREEN
6	GREEN	14	LIGHT BLUE
7	BLUE	15	DARK BLUE
8	YELLOW	16	LIGHT GREEN

In short, the COLOR command looks like this:

COLORarea, color, luminance

Here is a quick program to show you all the Plus/4's colors:

First type NEW and hit **RETURN**. Don't forget to hit
RETURN after each program line.

```
10 COLOR 0, 7, 7
20 FOR M= 0 TO 7
30 FOR N= 1 TO 2
40 FOR L= 1 TO 16
50 PRINT " " ; ; ;
60 READ A
70 COLOR 1, A, M
80 PRINT " " ;
90 NEXT L
100 PRINT
110 RESTORE
120 NEXT N,M
130 COLOR 1, 2, 4
200 DATA 7,14,4,13,6,16,11,8,10,9,3,12,5,15,2,1
```

Now type RUN and hit **RETURN**, to see a bright blue screen with each of the other 15 colors shown at each luminance level.

NOTE: Like most of the BASIC graphic terms reviewed in this chapter, COLOR may be referred to as a statement or command interchangeably. The distinction is unimportant in explaining COLOR, or any other graphic commands, since it is based on whether the term is more often used in direct or programming mode.

HIGH RESOLUTION GRAPHICS

Your Plus/4 screen contains 25 rows of 40 characters each, or 1000 total character positions on the screen. Each character is formed out of single dots, with 8 rows of 8 dots each making an entire character. Your screen has a total of 320 dots on each row, and 200 rows of dots, or 64,000 dots all together. Your Plus/4 has control over every single dot.

Using normal graphics, you have limited control over the individual dots. You must use the 256 characters in each character set built into the Plus/4, which lets you create many pictures. But think of how many you could create if you could control each dot by itself!

The high resolution graphics ability of the Plus/4 lets you do just that. You can use commands that let you draw and erase dots, lines, circles, and other shapes.

There is one limit to high-res graphics. The Plus/4 can only use two colors in each 8×8 character position. That is, each 8×8 space on the screen where characters usually go is limited to two colors (foreground and background color for that square). You can use different colors for each different character position, but only two colors within that position. Another graphic mode that will be covered later in this section, multi-color mode, allows up to four different colors per character position at the cost of the resolution available in the high-resolution mode.

This program utilizes some of the high resolution graphics capability of the Plus/4, in particular the GRAPHIC command. Start by typing NEW and hitting **RETURN**. Hit the **RETURN** key after typing each line. After typing in the entire program, type RUN and hit **RETURN** as usual.

```
10 COLOR 0,1
20 GRAPHIC 1,1
30 FOR L= 2 TO 16
40 COLOR 1,L,2
50 DRAW 1,0,L* 12 TO 319,L* 12
60 DRAW 1,L* 18,0 TO L* 18,199
70 NEXT L
80 FOR L= 1 TO 5000:NEXT
90 COLOR 1,2,3
100 GRAPHIC 0
```

Notice that the colors change near the intersections.

To switch from normal graphics (also called Text Mode) to high-res, just type the command GRAPHIC 2,1 and hit **RETURN**. What happens? The screen goes blank and the cursor reappears near the bottom of the screen. The Plus/4 divided up the screen into 2 separate sections: the top for graphics and the bottom five lines for text. If you don't

want the bottom five lines for text, you can use the command GRAPHIC 1,1, but you won't be able to see any commands you type.

You can switch back and forth from graphics to text using the GRAPHIC command. The command GRAPHIC 0 switches the screen back to text, while GRAPHIC 2 switches back to high-res without erasing the screen. Adding ,1 after the command erases the screen.

In general the GRAPHIC command looks like this:

GRAPHIC mode, clear ← this part is optional

Mode Number	Effect
0	Text
1	High-res
2	High-res + text
3	Multi-color
4	Multi-color + text

Clear Number	Effect
0	Don't clear screen
1	Clear screen

There is another way to clear the high-resolution screen. The command SCNCLR erases the screen without changing the graphic mode.

Once you use high-resolution graphics, the computer sets aside 10K of memory for your hi-res screen. This memory is taken from the BASIC program area. When you are through using graphics, you can reclaim this memory by using the command GRAPHIC CLR.

POINTS, LINES AND LABELS

Type the commands GRAPHIC 2,1: DRAW 1,0,0 and hit **RETURN**. Look closely at the upper left corner of the screen. The Plus/4 drew a black dot there.

In the DRAW command, the first number is either 1 (foreground color) or 0 (background color). The next two numbers are for the row and column positions for the dot. So if you wanted to draw a dot at column 17, row 20, just type DRAW 0,17,20. To erase the same dot type DRAW 0,17,20.

The DRAW command can also draw a line between any two points. Just add the word TO and the coordinates of the

other end, like this: DRAW 1,1,1 TO 100,100. This draws a line from 1,1 to 100,100.

If you are used to drawing graphs in math, you might get a little confused at first while using the computer. The coordinate system in the Plus/4 is different from what you're used to. In math the 0,0 point would either be at the center or the lower left corner of the screen, but on the computer it is the upper left corner. You'll get used to the system in the computer as you practice.

Once you have put a point or line on the screen, you can draw a line from it to any other point like this: DRAW 1 TO 150,50. This draws a line from the last point drawn to column 150, row 50. If your program uses a lot of DRAW TO commands, you could place the first dot at a position on the screen by using the LOCATE command, as in LOCATE 100,100.

The DRAW command can have several forms, such as:


COMMAND	RESULT
DRAW color source, column, row	POINT
DRAW color source, column, row TO column, row	LINE
DRAW color source TO column, row	LINE DRAWN FROM LAST POINT

Color source is 0 for the background, 1 for the foreground

To erase points or lines on the screen, use the DRAW command followed by the number 0. If you created a point with DRAW 1,1,1, you can erase it with DRAW 0,1,1. A line created with DRAW 1,1,1 TO 100,100 is erased by DRAW 0,1,1 TO 100,100.

This program draws a curve based on the sine function. Type NEW and hit . Remember to hit the key after each line, then type RUN.

```
10 COLOR 0,1
20 COLOR 1,2
30 GRAPHIC 1,1
40 LOCATE 0,100
50 FOR X= 1 TO 319
60 Y = INT (100+ 99* SIN(X/20))
70 DRAW 1 TO X,Y
80 NEXT X
90 FOR L= 1 TO 5000
100 NEXT L
110 GRAPHIC 0
```



Don't type NEW after RUNning the last program. To plot the program differently, change line 70 to:

70 DRAW 1, X, Y


This program plots the same curve using points instead of lines.

The Char Command

Graphs are more easily understood and useful if you label them. You can use the CHAR command to mix text right into a high resolution drawing. For instance, the statement CHAR 1,0,5,"HELLO" puts the word HELLO into the fifth row at the left edge of the screen. The first number after the word CHAR is either 1 (for draw) or 0 (for erase). The next two numbers are the column and row where the text appears.

Leave the last two programs in the computer: don't type NEW. Add these lines:

```
81 CHAR 1,0,0,"GRAPH OF":CHAR 1,0,1,"FORMULA"  
82 CHAR 1,0,2,"Y= SIN(X)"  
83 DRAW 1,0,100 TO 319,100,189,0 TO 189,199  
84 CHAR 1,0,12,"X-AXIS": CHAR 1,22,0,"Y"  
85 CHAR 1,22,2,"A": CHAR 1,22,3,"X"  
86 CHAR 1,22,4,"I": CHAR 1,22,5,"S"
```



SQUARES, CIRCLES, POLYGONS AND PAINTING

Drawing Rectangles

Using the DRAW command, you can draw pictures by plotting many dots or lines. To draw a square, you can use the command DRAW 1,0,0 TO 100,0 TO 100,100 TO 0,100 TO 0,0 (plotting each endpoint of the square) or you can just use the BOX command.

THE BOX COMMAND

Your Plus/4 includes a command to make it easier to draw squares and other rectangular shapes. The BOX command lets you pick the points of 2 opposite corners of the square. To duplicate the same box as in the above example, just use BOX 1,0,0,100,100. The number 1 again means that you want to draw and not erase. The next four numbers are the coordinates of the box's opposite corners, (0,0) at the upper-left corner and (100,100) near the middle of the screen.

The BOX command can form any rectangle just by changing the corners. You can even rotate the box by specifying an angle (in degrees) after the last coordinate, like this: BOX 1,50,50,100,100,45. The box is rotated 45 degrees clockwise.

If you would like to draw a solid box instead of just the outline, you just add a comma 1 after the angle. A solid box at the center of the screen is shown as BOX 1,100,50,220,150,,1. Notice that you need a comma to take the place for the angle, even though you don't want the box rotated.

Some typical forms of the BOX command are:

COMMAND	EFFECT
BOXon, column1, row1, column2, row2	Outline
BOXon, col1, row1, col2, row2, angle	Rotated
BOXon, col1, row1, col2, row2, , fill	Solid box
BOXoff, col1, row1, col2, row2, angle, fill	Erase area of screen

Here are a couple of programs that use the BOX command:

Don't forget to type NEW then hit **RETURN** before entering each program, and press **RETURN** after typing in each line.

```
10 COLOR 0,1
20 COLOR 1,2
30 GRAPHIC 2,1
40 A = RND(1)* 20+ 10
50 FOR L= 0 TO 359 STEP A
60 BOX 1, 100, 30, 220, 130, L
70 NEXT L
80 FOR L= 1 TO 2000: NEXT L
90 GRAPHIC 0,1
```

```
5 TRAP 60
10 GRAPHIC 2,1
20 DEF FNA(X)= INT( RND(1)* X)
30 COLOR 1, FNA(15)+ 1
40 BOX 1, FNA(320), FNA(160), FNA(320), FNA(160),, 1
50 GOTO 30
60 COLOR 1,2,3: GRAPHIC 0
```

Hit **RETURN** and type RUN after each program is completely entered. Hold down the **STOP** key to end the program.

The second program draws different colored squares all over the screen. You'll notice some parts of the screen changing when other parts near them change. The reason for this was discussed earlier in this chapter.

Drawing Circles

Your Plus/4 also has commands for drawing circles. Like the BOX command, we can vary the shape of the circle to form an oval (also called an ellipse), and we can rotate the oval. We can also just draw a section of the shape (called an arc).

This command draws a circle in the center of the screen: CIRCLE 1,160,100,50. This tells the Plus/4 to draw a circle with its center at row 160 and column 100, with a radius of 50. This actually produces an oval, since the dots on the screen are taller than they are wide. To change this to a real circle you must add a separate number to tell that the height is different from the width, like this: CIRCLE 1,160,100,50,42.

The Plus/4 can also draw a square, triangle or other polygon using the CIRCLE command. Just tell the computer how many degrees to go between points on the circle, like this: CIRCLE 1,160,100,50,42,,,120. This command draws a triangle, since each side is 120 degrees. (Omitting number values while including commas in a graphic command causes the computer to read standard default values for the missing number.) A simple formula to get the angle for a polygon with N sides is $360/N$.

Here's a quick program for drawing polygons:

```
10 GRAPHIC 2,1
20 INPUT "HOW MANY SIDES";A
30 IF A<2 OR A>100 THEN PRINT "DON'T BE
RIDICULOUS": GOTO 20
40 CIRCLE 1,160,80,40,33,,,360/A
50 GOTO 20
```

You can choose to draw only an arc instead of a whole circle. The CIRCLE command accepts the starting and ending angles in degrees, right after the height number. The command

CIRCLE 1,160,100,50,42,90,180 displays only the lower right section of the circle.

To rotate an oval, add the angle of clockwise rotation after the command, like this example: CIRCLE 1,160,100,100,20,,,30.

The usual forms of the CIRCLE command are:

COMMAND	EFFECT
CIRCLE on, center column, center row, radius	oval
CIRCLE on, c-col, c-row, width, height	circle/oval
CIRCLE on, c-col, c-row, wid, ht, start, finish	arc
CIRCLE on, c-col, c-row, width, height,,,angle	rotate oval
CIRCLE on, c-col, c-row, wid, ht,,,point angle	polygon

NOTE: When there are commas without numbers in a command such as CIRCLE or BOX, the Plus/4 automatically interprets the comma as an input of the default value for that parameter of the command. For example, CIRCLE,160,40,100,100 is read by the computer as CIRCLE 1, 160..., reading the default value of 1 for the color source.


This next program uses the CIRCLE command for an interesting effect. Type NEW then hit ~~EXECUTE~~ to erase the last program from memory before typing in this program.

```
10 COLOR 0,1  
20 COLOR 1,2  
30 GRAPHIC 1,1  
40 A = RND(1)* 20+ 10  
50 FOR L= 0 TO 359 STEP A  
60 CIRCLE 1, 160, 100, 80, 40,,,L  
70 NEXT L  
80 FOR L= 1 TO 2000: NEXT L  
90 GRAPHIC 0,1
```

Here's a program you can try that uses the CIRCLE command to create a simple design.

```
NEW  
10 COLOR 0,1  
20 COLOR 1,2  
30 GRAPHIC 2,1  
40 FOR L= 1 TO 4  
50 Y= 50  
60 IF L= 2 OR L= 4 THEN Y= 100  
70 X= L* 35 + 50  
80 CIRCLE 1, X, Y, 50, 42
```

```
90 NEXT L  
100 PRINT "PLUS/4 CIRCLES"
```



The Paint Command

The PAINT command fills in any enclosed area up to the boundaries formed by any lines drawn on the screen. If there are no drawn lines, the screen is filled right to the edge. The BOX command can be used to fill in boxes and rectangles with color. The PAINT command can color in irregular shapes and other non-uniform areas on the screen that can't be filled with other commands.

In the last exercise, we created a 4-ring symbol. By adding some PAINT commands to the program we can color in only the areas between the rings.

Add these lines to the last program:

```
110 FOR L= 0 TO 2  
120 PAINT 1,120 + 35 *L, 75  
130 NEXT L
```

MULTI-COLOR GRAPHICS

The Plus/4 high resolution graphics give you control over every single dot or "pixel" on the screen, but you have seen that the ability to put colors close together is limited. Most high-res programs can use only one or two colors.

For including more different colors, your Plus/4 has a special "in-between" graphics mode called multi-color graphics. In multi-color graphics, you control half as many dots on each row as in high-res because each dot is twice as wide. You get 160 dots on each row, while still getting 200 rows. There is a trade-off for the use of multiple colors, which is slightly lower resolution.

To begin using multi-color graphics, review the GRAPHIC command earlier in this chapter. You'll see that the multi-color screen without text is GRAPHIC 3 and the multi-color screen with 5 lines of text is GRAPHIC 4.

Now look at the table listing the COLOR command. There are two areas that we haven't used yet, areas 2 and 3. These

areas hold two extra colors. You can use any of the three colors (1, the text color; 2, an extra color; and 3, another extra color). These colors do not interfere with each other on the screen the way the high-res colors do in some previous programs in this chapter.

These two programs make use of multi-color graphics, the first with the rings and the second showing a "neon sign" effect.

```
10 COLOR 0,1
20 GRAPHIC 4,1
30 FOR L= 1 TO 4
40 Q=L: IF Q>3 THEN Q= Q-3
50 COLOR Q,L+1
60 Y= 50
70 IF L= 2 OR L= 4 THEN Y= 100
80 X= L*18 + 25
90 CIRCLE Q, X, Y, 25, 42
100 NEXT L
TYPE NEW then HIT RETURN. Don't forget to hit the RETURN key after each line. Type RUN and hit RETURN at the end.
```

```
10 COLOR 0,1
20 GRAPHIC 3,1
30 COLOR 3,1
40 TRAP 200
50 DRAW 3,10,10 TO 10,100: DRAW 3,10,55 TO 30,55
60 DRAW 3,30,10 TO 30,100: DRAW 3,50,10 TO 80,10
70 DRAW 3,65,10 TO 65,100: DRAW 3,50,100 TO 80,100
80 FOR L= 0 TO 7
90 COLOR 3,2,L
100 FOR M= 1 TO 100: NEXT M
110 NEXT L
120 COLOR 3,1
130 FOR M= 1 TO 100: NEXT M
140 GOTO 80
200 GRAPHIC 0: COLOR 1,2,7
```

Color area 3, the second of the multicolor areas, has a special ability that none of the others has. Once you have drawn on the screen using area 3, you can change that color *everywhere* it appears on the screen by using the COLOR command. If you set the color with COLOR 3,5 and draw using that color, your graphics appear in purple. If you then change the color with COLOR 3,6, all the purple areas would change to green. This doesn't work with any other area.

The Plus/4 Programmer's Reference Guide contains more information about graphics.
